Appendix A2. Picard Iteration

The following first order initial value problem

$$y'(x) = f(\, x \,,\, y(x)\,) \;\;,\;\; y(\text{x0}) = \text{y0}$$

can be reformulated as the an integral equation

$$y(t) = \text{y0} + \int_{\text{x0}}^{x} f(t,\, y(t))\, dt$$

See Simmons/Krantz, Chapter 3, Section 2.2.

Assume that $f(x, y)$ and its y partial are continuous on a rectangle containing the point (x0,y0). Then starting with an initial "guess" for a solution

$$y = y_0(x)$$

the recursive relation

$$y_{j+1}(t) = \text{y0} + \int_{\text{x0}}^{x} f(t,\, y_j(t))\, dt$$

generates a sequence of functions converging to the solution. These functions are called the Picard iterates generated by $y_0(x)$.

Mathematica can generate some of these iterates, use an iteratively defined process.

The following example is Example 3.16 in Simmons/Krantz.

<u>Example</u> Obtain the first three Picard iterates for the following IVP, generated by the function $g(t) = 0$.

$$y' = 2y,\;\; y(0) = 1$$

Sketch their graphs and the graph of the exact solution.

Start with the definition of the function f.

```
In[66]:= f[x_,y_] := 2*y
```

Use it to define the differential equation.

```
In[67]:= DE = y'[x] == f[x,y[x]]
Out[67]= y′[x] == 2 y[x]
```

Define the expression Y[0,x] to be 1. Then define Y[n,x] recursively as shown below.

```
In[71]:= Y[0,x_] := 1
         Y[n_,x_] := Y[0,x] + Integrate[f[t,Y[n-1,t]], {t,0,x}]
```

Here are the first three Picard iterates.

In[74]:=
```
Table[Y[n,x], {n,1,3} ]
```

Out[74]= $\left\{1+2x,\ 1+2x+2x^2,\ 1+2x+2x^2+\frac{4x^3}{3}\right\}$

The solution formula is obtained next.

In[76]:=
```
soln = DSolve[ {DE, y[0]==1}, y[x], x ]
```

Out[76]= $\{\{y[x] \to e^{2x}\}\}$

The plot of the solution, red and thick, and the three iterates, blue, green, black:

In[77]:=
```
Show[ Plot[ y[x]/.soln, {x,0,2},
            PlotStyle->{RGBColor[1,0,0],Thickness[0.02]} ],
      Plot[ Y[1,x], {x,0,2}, PlotStyle->{RGBColor[0,0,1]} ],
      Plot[ Y[2,x], {x,0,2}, PlotStyle->{RGBColor[0,1,0]} ],
      Plot[ Y[3,x], {x,0,2} ], PlotRange->{0,5} ]
```



In the next output **Series** is used to generate a Taylor series approximation to the solution. Note that the Picard iterates above are, in fact, the Taylor series approximations to the solution.

In[80]:=
```
Series[ y[x]/.soln[[1]], {x,0,6} ];
Normal[%]
```

Out[81]= $1+2x+2x^2+\frac{4x^3}{3}+\frac{2x^4}{3}+\frac{4x^5}{15}+\frac{4x^6}{45}$